HEWLETT PACKARD

Effective Multi-tap Transformer Testing Using Scanner

Application Note 1224-5

HP 4263A LCR Meter





Introduction

With the progress of recent electronics equipment and digital networks, production amounts are increasing of the transformers which contribute to equipment miniaturization, low power dissipation and higher quality. Therefore, improvement of select estimate efficiency is required at the production line or incoming inspection. Noticed recently, improvement of estimation efficiency is required for pulse transformers which are used in LAN or ISDN digital networks, and for multi-tap transformers with three or more pole taps, such as switching power transformers. This application note shows an effective multi-tap transformer measurement using a scanner and the HP 4263A LCR Meter.

HP 4263A Transformer Measurement Capability

The HP 4263A LCR Meter is a low price instrument which measures the fundamental parameters of LCR components with speeds as fast as 25ms, at frequencies of 100, 120, 1k, 10k and 100kHz. In addition, with option 001, the HP 4263A measures turns ratio (N), mutual inductance (M) and d.c.resistance (DCR) which are required for transformer measurement. Figure 1 shows a HP 4263A simple block diagram for L, M, and DCR measurement.



Figure 1. HP 4263A Block Diagram for L, M, and DCR measurement

For example, in the inductance-turns ratio (L-N) measurement, an AC voltage is applied at the Hcur terminal. Self-inductance value (L1) is calculated from the measured values of V1 and I1. Turns ratio (N) is automatically obtained from the ratio of measured values V1 and V2 (discriminating the polarity simultaneously).

In the d.c. resistance (of L-DCR) measurement, the applied voltage at the Hcur terminal is DC. D.C. resistance value (DCR1) is calculated from the measured values V1 and I1.

There are, however, the following limitations when using the measurement connection.

- Only primary self-inductance and d.c. resistance of the transformer can be measured. For the secondary values, the transformer connections must be changed.
- Turns ratio must be 0.9 or more (In the case of less than 0.9, the measurement is not performed due to saturation of internal circuitry).

HP 16060A Transformer Test Fixture can be used to overcome these limitations. By changing the external switch of this fixture, connections to the transformer are changed and thus both primary and secondary parameters and turns ratio can easily be measured. Figure 2 shows the simple block diagram of the HP 16060A.



Figure 2. HP 16060A Block Diagram

Multi-tap Transformer Measurement Using a Scanner

Multi-tap Transformers having two or more poles can be measured with the HP 4263A and a scanner.

(A) System Configuration

Figure 3 shows the system configuration for measuring a multi-tap transformer that has 4 taps.



Figure 3. System Configuration for multi-tap transformer

The HP 3488A switch/control unit with a 4 x 4 matrix switch module (option 013) is used. Option 013 offers highly flexible switching, and any combination of 4 input channels may be connected to any combination of 4 output channels. Thus option 013 is suitable for testing the multi-tap transformer. Figure 4 shows the hardware configuration of the 4 x 4 matrix switch module.



Figure 4. Option 013 4 x 4 Matrix Module

(B) System Construction Recommendations

When constructing the system, the following points must be considered to assure the measurements are as precise as can be. (refer to figure 5)

1. Make measurement cables as short as possible. The parasitic inductance and resistance of measurement cables make a large contribution to measurement error. For recommendable length, conductive wire inductance value must be 1/10 or less than the measured inductance value (similarly conductive wire resistance).

2. Configure into a shielded 2 terminal configuration, to prevent the influence of external noise or stray capacitance.

3. Connect the low terminals close to the transformer. In the HP 4263A transformer measurement, the primary and secondary inductors' low terminals of the transformer must be connected together. When using a scanner, these connections should be close to the transformer under test. If connecting at a far point from the transformer (for example, input point of scanner module), low side wire resistance would contribute to increase measurement error.



Figure 5. System Construction

(C) Measurement Procedure

All measurements of the multi-tap transformer, self-inductance, d.c. resistance, and turns ratio, can be measured with only one connection by using the sample program shown at the end of this note (for HP 9000 Series 300 Controller). Figure 6 shows the flow chart of the sample program.



Figure 6. Flow Chart of Sample Program

This program executes the open and short corrections and displays each measured value of each tap of the transformer. If turns ratio measurement cannot be made due to the condition that turns ratio must be 0.9 or greater, the scanner will be automatically changed and the measurements re-done. This program can be modified to match other systems or conditions.

The following steps outline the program procedure:

STEP1. Run the program. The following message is displayed on the controller's display.

```
SELECTION (1) MEASUREMENT (2) CORRECTION ?

TYPE NUMBER AND PRESS RETURN KEY
```

At this point, select the measurement directly, or first the measurement of correction data. To execute the measurement, type 1 and press RETURN key on the controller (Go to Step 4). To measure the correction data, type 2 and press RETURN key on the controller.

STEP2. If the measurement of correction data in step 1 was selected, the following message is displayed on the controller's display. The OPEN correction data of each channel of the scanner (CH.0-CH.3) is now measured.

```
CH.0 OPEN MEASUREMENT
OPEN TEST TERMINALS OF CH.0
(1) START OPEN MEAS.  (2) SKIP CH.0 OPEN MEAS?

TYPE NUMBER AND PRESS RETURN KEY
```

To measure the OPEN correction data, set all channels to the OPEN condition as shown in Figure 7. Then, type 1 and press RETURN key on the controller. OPEN correction data of channel number 0 (CH.0) is acquired. Continue to acquire data for channels 1 - 3.



Figure 7. OPEN Condition

STEP3. After the OPEN correction measurements are completed, the following messages is displayed on the controller's display. The SHORT correction data of each channel of the scanner (CH.0 - CH.3) is now measured.

```
CH.0 SHORT MEASUREMENT
SHORT TEST TERMINALS OF CH.0
(1) START SHORT MEAS.  (2) SKIP CH.0 SHORT MEAS.?

TYPE NUMBER AND PRESS RETURN KEY
```

To measure the SHORT correction data, set all channels to SHORT condition as shown in Figure 8. Then, type 1 and press the RETURN key on the controller. SHORT correction data of channel number 0 (CH.0) is acquired. Continue to acquire data for channels 1 - 3.



Figure 8. SHORT Condition

STEP4. After the OPEN/SHORT correction data is acquired, the following message (same as in STEP1) is displayed on the controller's display.

```
SELECTION (1) MEASUREMENT (2) CORRECTION?

TYPE NUMBER AND PRESS RETURN KEY
```

To execute the measurement, connect the multi-tap transformer under test to the scanner as shown in Figure 9. Type 1 and press the RETURN key on the controller.



Figure 9. Connection of Multi-tap Transformer

Self-inductance, d.c. resistance and turns ratio are measured by scanning each tap of the multi-tap transformer.

```
N1: L[H]: 6.00928E-6    DCR [OHM]: .0134568726173   N: 1
N2: L[H]: 2.392557E-5   DCR [OHM]: .0171348134407   N: 2.1304
N3: L[H]: 9.603832E-5   DCR [OHM]: .0230939715609   N: 4.0630
N4: L[H]: .00038334126  DCR [OHM]: .0250939715609   N: 8.0188

DO YOU WANT TO CONTINUE TO MEASURE (1) YES (2)NO
```

STEP5. If you want to repeat the measurement, type 1 and press RETURN key on the controller. Or to end the program, type 2 and press RETURN key on the controller.

(D) Additional Measurement Error

The system configuration shown in Figure 3, slightly increases measurement errors, in comparison with measured values using the HP 16060A Transformer Test Fixture. These errors (supplemental characteristics) are the following using frequency: 1 kHz, signal level: 1 Vrms, measurement time: Medium.

Self-inductance: Refer to Figure 10
D.C. Resistance: Refer to Figure 11
Turns Ratio: 0.02 % or less



Figure 10. Self-Inductance Additional Error



Figure 11. D.C. Resistance Additional Error

Conclusion

By combining the HP 4263A (with Option 001) with a scanner, the required parameters of a multi-tap transformer can be measured with only one connection. Using this method improves efficiency at the production line or incoming inspection.

Appendix. Sample Program

```
10    !*****************************************
20    !* HP 4263A with Option 001              *
30    !*  Transformer Measurement using scanner *
40    !*****************************************
50    !
60    OPTION BASE 0
70    DIM Ch_hc(3),Ch_hp(3)
80    DIM Meas_r(3),Meas_l(3),Dummy(3),N(3),True_r(3),True_l(3)
90    DIM Open_r(3),Open_l(3),Open_g(3),Open_b(3),Short_r(3),Short_l(3)
100                                          !
110   Hp4263a=717                            ! HP 4263A HP-IB Address = 717
120   Hp3488a=709                            ! HP 3488A HP-IB Address = 709
130   Nch=3                                  ! (#-1) of Transformer tap
140   F=1.0E+3                               ! Test Frequency
150   V=1                                    ! Test Signal Level
160   T=.065                                 ! Measurement Speed
170   N(0)=1                                 ! N1=1 as reference
180                                          !
190 Main_menu:                               ! << MAIN MENU >>
200                                          !
210   PRINT CHR$(12)                         ! Clear screen
220   Work=0                                 !
230   PRINT "SELECT FUNCTION (1) MEASUREMENT (2) CORRECTION ?" !
240   INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work !
250   IF Work=1 THEN Measurement             !
260   IF Work=2 THEN Correction              !
270                                          !
280 Correction:                              ! << CORRECTION >>
290                                          !
300 Open_correction:                         ! << OPEN correction >>
310                                          !
320   OUTPUT Hp4263a;":SYSTEM:PRESET"        ! Reset the HP 4263A
330   OUTPUT Hp4263a;":SOURCE:FREQ ";F       !  Frequency: F
340   OUTPUT Hp4263a;":SOURCE:VOLTAGE ";V    !  Signal level: V
350   OUTPUT Hp4263a;":SENS:FIMP:APER 0.5"   !  Meas. speed: LONG
360                                          !
370   FOR Ch=0 TO Nch                        !
380                                          !
390     Ch$=VAL$(Ch)                         !
400     PRINT CHR$(12)                       !
410     PRINT "CH."&Ch$&" OPEN MEASUREMENT"  !
420     PRINT "OPEN TEST TERMINALS OF CH."&Ch$ !
430     PRINT " (1) START OPEN MEAS.    (2) SKIP CH."&Ch$&" OPEN MEAS.?"
440     Work=0                               !
450     INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work !
460     IF Work<>1 AND Work<>2 THEN 450      !
470     IF Work=1 THEN Open_meas             !
480     IF Work=2 THEN Open_skip_ch          !
490                                          !
500 Open_meas:                               !
510                                          !
520     Ch_hc(Ch)=200+Ch*10                  ! Channel Setting of Hcur/Lcur
530     IF Ch=0 THEN Ch_hp(Ch)=211           ! Channel Setting of Hpot/Lpot
540     IF Ch<>0 THEN Ch_hp(Ch)=201          !
550                                          !
560     OUTPUT Hp3488a;"RESET"               ! Reset the HP 3488A
570     OUTPUT Hp3488a;"CLOSE";Ch_hc(Ch),Ch_hp(Ch) ! Close the channels
580     OUTPUT Hp4263a;":SENS:FUNC:CONC ON"  !  Meas.mode: L2-R2
590     OUTPUT Hp4263a;":SENS:FUNC 'IMP','RES'"!
600     OUTPUT Hp4263a;":CALC1:FORM LS"      !
610     OUTPUT Hp4263a;":CALC2:FORM REAL"    !
620     OUTPUT Hp4263a;"TRIG:SOUR BUS"       !  Trigger mode: BUS
630     OUTPUT Hp4263a;"*TRG"                ! OPEN correction data
640     ENTER Hp4263a;S,Open_l(Ch),Open_r(Ch)  !
650     IF S<>0 THEN 630
660     OUTPUT Hp3488a;"OPEN";Ch_hc(Ch),Ch_hp(Ch) ! Open the channels
670     Open_g(Ch)=1/Open_r(Ch)              !
680     Open_b(Ch)=1/Open_l(Ch)              !
690                                          !
700 Open_skip_ch:                            !
710                                          !
720   NEXT Ch                                !
730                                          !
740 Short_correct:                           ! << SHORT Correction >>
750                                          !
760   OUTPUT Hp4263a;":SYSTEM:PRESET"        ! Reset the HP 4263A
770   OUTPUT Hp4263a;":SOURCE:FREQ ";F       !  Frequency: F
780   OUTPUT Hp4263a;":SOURCE:VOLTAGE ";V    !  Signal level: V
790   OUTPUT Hp4263a;":SENS:FIMP:APER 0.5"   !  Meas. speed: LONG
800                                          !
810   FOR Ch=0 TO Nch                        !
820                                          !
830     PRINT CHR$(12)                       ! Clear screen
840     Ch$=VAL$(Ch)                         !
850     PRINT "CH."&Ch$&" SHORT MEASUREMENT" !
860     PRINT "SHORT TEST TERMINALS OF CH."&Ch$!
870     PRINT "(1) START SHORT MEAS.   (2) SKIP CH."&Ch$&" SHORT MEAS.?" !
880     Work=0                               !
890     INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work !
900     IF Work<>1 AND Work<>2 THEN 890      !
910     IF Work=1 THEN Short_meas            !
920     IF Work=2 THEN Short_skip_ch         !
930                                          !
940 Short_meas:                              !
950                                          !
```

```
960     Ch_hc(Ch)=200+Ch*10                     ! Channel Setting for Hcur/Lcur
970     IF Ch=0 THEN Ch_hp(Ch)=211              ! Channel Setting for Hpot/Lpot
980     IF Ch<>0 THEN Ch_hp(Ch)=201             !
990                                             !
1000    OUTPUT Hp3488a;"RESET"                  ! Reset the HP 3488A
1010    OUTPUT Hp3488a;"CLOSE";Ch_hc(Ch),Ch_hp(Ch) ! Close the channels
1020    OUTPUT Hp4263a;":SENS:FUNC:CONC ON"     ! Meas.mode: L2-R2
1030    OUTPUT Hp4263a;":SENS:FUNC 'IMP','RES'" !
1040    OUTPUT Hp4263a;":CALC1:FORM LS"         !
1050    OUTPUT Hp4263a;":CALC2:FORM REAL"       !
1060    OUTPUT Hp4263a;"TRIG:SOUR BUS"          ! Trigger mode: BUS
1070    OUTPUT Hp4263a;"*TRG"                   ! SHORT correction data
1080    ENTER Hp4263a;S,Short_l(Ch),Short_r(Ch) !
1090    IF S<>0 THEN 1070                       !
1100    OUTPUT Hp3488a;"OPEN";Ch_hc(Ch),Ch_hp(Ch) ! Open the channels
1110                                            !
1120 Short_skip_ch:                             !
1130                                            !
1140  NEXT Ch                                   !
1150                                            !
1160  GOTO Main_menu                            ! Return to Main Menu
1170                                            !
1180 Measurement:                               ! << MEASUREMENT >>
1190                                            !
1200  PRINT CHR$(12)                            ! Clear screen
1210  OUTPUT Hp4263a;":SYSTEM:PRESET"           ! Reset the HP 4263A
1220  OUTPUT Hp4263a;":SOURCE:FREQ ";F          !  Frequency: F
1230  OUTPUT Hp4263a;":SOURCE:VOLTAGE ";V       !  Teset signal level: V
1240  OUTPUT Hp4263a;":SENS:FIMP:APER ";T       !  Measurement Speed: T
1250  OUTPUT Hp4263a;":TRIG:SOUR BUS"           !  Trigger mode: BUS
1260  OUTPUT Hp3488a;"RESET"                    ! Reset the HP 3488A
1270                                            !
1280  FOR Ch=0 TO Nch                           !
1290                                            !
1300    OUTPUT Hp4263a;":SENS:FUNC:CONC ON"     !  Meas.mode: L2-R2
1310    OUTPUT Hp4263a;":SENS:FUNC 'IMP','RES'" !
1320    OUTPUT Hp4263a;":CALC1:FORM LS"         !
1330    OUTPUT Hp4263a;":CALC2:FORM REAL"       !
1340    Ch_hc(Ch)=200+Ch*10                     ! Channel Setting for Hcur/Lcur
1350    IF Ch=0 THEN Ch_hp(Ch)=211              ! Channel Setting for Hpot/Lpot
1360    IF Ch<>0 THEN Ch_hp(Ch)=201             !
1370    OUTPUT Hp3488a;"CLOSE";Ch_hc(Ch),Ch_hp(Ch) ! Close the channels
1380    OUTPUT Hp4263a;"*TRG"                   ! L2-R2 measurement
1390    ENTER Hp4263a;S,Meas_l(Ch),Meas_r(Ch)   !
1400    True_l(Ch)=(Meas_l(Ch)-Short_l(Ch))/((1-(Meas_l(Ch)-Short_l(Ch))*Open_b(Ch))) !
1410    True_r(Ch)=(Meas_r(Ch)-Short_r(Ch))/((1-(Meas_r(Ch)-Short_r(Ch))*Open_g(Ch))) !
1420                                            !
1430    IF Ch=0 THEN Skip_meas                  !
1440                                            !
1450    OUTPUT Hp4263a;":SENS:FUNC 'IMP','VOLT:AC'" ! Meas.mode: L-N
1460    OUTPUT Hp4263a;"*TRG"                   ! N measurement
1470    ENTER 717;S,Dummy(Ch),N(Ch)             !
1480                                            !
1490    IF S=1 THEN                             ! Hcur-Hpot CHANGE for OVLD
1500      OUTPUT Hp3488a;"OPEN";Ch_hc(Ch),Ch_hp(Ch) ! Open channels
1510      Ch_hc(Ch)=201+Ch*10                   ! Hcur channel CHANGE
1520      Ch_hp(Ch)=200                         ! Hpot channel CHANGE
1530      OUTPUT Hp3488a;"CLOSE";Ch_hc(Ch),Ch_hp(Ch) ! Close the changed ch.
1540      OUTPUT Hp4263a;"*TRG"                 ! N measurement
1550      ENTER Hp4263a;S,Dummy(Ch),N(Ch)       !
1560      N(Ch)=N(0)/N(Ch)                      !  N1:Nx=1:0.XX
1570    END IF                                  !
1580                                            !
1590 Skip_meas:                                 !
1600                                            !
1610    PRINT "N";Ch+1;":","L [H]:";True_l(Ch),"DCR [OHM]:";True_r(Ch),"N:";N(Ch)
1620    OUTPUT Hp3488a;"OPEN";Ch_hc(Ch),Ch_hp(Ch)! Open channels
1630                                            !
1640  NEXT Ch                                   !
1650                                            !
1660  Work=0                                    !
1670  INPUT "DO YOU WANT TO CONTINUE TO MEASURE?   (1) YES  (2) NO",Work
1680  IF Work=1 THEN Measurement                !
1690  IF Work=2 THEN 1720                       !
1700  IF Work<>1 AND Work<>2 THEN 1670          !
1710                                            !
1720  END
```

HEWLETT PACKARD

For more informatin call your local HP sales office listed in your telephone directory or an HP regional office listed below for the location of your nearest sales office.

United States:
Hewlett-Packard Company
4 Choke Cherry Road
Rockville, MD 20850

Hewlett-Packard Company
5201 Tollview Drive
Rolling Meadows, IL 60008

Hewlett-Packard Company
5161 Lankershim Blvd.
No. Hollywood, CA 91601

Hewlett-Packard Company
2015 South Park Place
Atlanta, GA 30339

Canada:
Hewlett-Packard Ltd.
6877 Goreway Drive
Mississauga, Ontario L4V1M8

Australia/New Zealand:
Hewlett-Packard Australia Ltd.
31-41 Joseph Street
Blackburn, Victoria 3130
Melbourne, Australia

Europe/Africa/Middle East:
Hewlett-Packard S.A.
Central Mailing Department
P.O. Box 529
1180 AM Amstelveen
The Netherlands

Far East:
Hewlett-Packard Asia Ltd.
22/F Bond Centre
West Tower
89 Queensway
Central, Hong Kong

Japan:
Yokogawa-Hewlett-Packard Ltd.
29-21, Takaido-Higashi 3-chome
Suginami-ku, Tokyo 168

Latin America:
Latin American Region Headquarters
Monte Pelvoux Nbr. 111
Lomas de Chapultepec
11000 Mexico, D.F. Mexico